
LABORATOR 4 

 
INSTRUCłIUNI SECVENłIALE ÎN LIMBAJUL VHDL 

Partea a 2-a 
 

 
 

In aceasta lucrare de laborator sunt prezentate o serie de instrucŃiuni secvenŃiale care 
pot apare într-un proces sau subprogram: instrucŃiunea case, instrucŃiunile de buclare (loop, 
while loop, for loop, next, exit) şi instrucŃiunea assert secvenŃială.  

 
4.1. InstrucŃiuni secvenŃiale 
4.1.1. InstrucŃiunea case 

 
Ca şi instrucŃiunea if, instrucŃiunea case selectează pentru execuŃie una din mai multe 

secvenŃe alternative de instrucŃiuni pe baza valorii unei expresii. Spre deosebire de 
instrucŃiunea if, în cazul instrucŃiunii case expresia nu trebuie să fie booleană, ci poate fi 
reprezentată de un semnal, o variabilă sau o expresie de orice tip discret (un tip enumerat sau 
întreg) sau un tablou unidimensional de caractere (inclusiv bit_vector sau std_logic_vector). 
InstrucŃiunea case se utilizează atunci când există un număr mare de alternative posibile. 

Această instrucŃiune este mai lizibilă decât o instrucŃiune if cu un număr mare de 
ramuri, permiŃând identificarea uşoară a unei valori şi a secvenŃei de instrucŃiuni asociate. 

Sintaxa instrucŃiunii case este următoarea: 
case expresie is 

when opŃiuni_1 => 
secvenŃă_de_instrucŃiuni 

... 
when opŃiuni_n => 

secvenŃă_de_instrucŃiuni 

[when others => 
secvenŃă_de_instrucŃiuni] 

end case; 
 

InstrucŃiunea case conŃine mai multe clauze when, fiecare specificând una sau mai 
multe opŃiuni. OpŃiunile reprezintă fie o valoare individuală, fie un set de valori cu care se 
compară expresia instrucŃiunii case. În cazul în care expresia este egală cu valoarea 
individuală sau cu una din valorile setului, se execută secvenŃa de instrucŃiuni specificată după 
simbolul =>. O secvenŃă de instrucŃiuni poate fi formată şi din instrucŃiunea nulă, null. Spre 
deosebire de anumite limbaje de programare, instrucŃiunile dintr-o secvenŃă nu trebuie incluse 
între cuvintele cheie begin şi end. Clauza others se poate utiliza pentru a specifica execuŃia 
unei secvenŃe de instrucŃiuni în cazul în care valoarea expresiei nu este egală cu nici una din 
valorile specificate în clauzele when. 

În cazul în care o opŃiune este reprezentată de un set de valori, se pot specifica fie 
valorile individuale din set, separate prin simbolul “|” având semnificaŃia “sau”, fie domeniul 
valorilor, fie o combinaŃie a acestora, după cum se arată în exemplul următor. 

case expresie is 
when val => 

secvenŃă_de_instrucŃiuni 

when val1 | val2 | ... | valn => 
secvenŃă_de_instrucŃiuni 

when val3 to val4 => 
secvenŃă_de_instrucŃiuni 

when val5 to val6 | val7 to val8 => 
secvenŃă_de_instrucŃiuni 

... 
when others => 

secvenŃă_de_instrucŃiuni 

end case; 
 

ObservaŃii 
• Într-o instrucŃiune case trebuie enumerate toate valorile posibile pe care le poate avea 

expresia de selecŃie, Ńinând cont de tipul sau subtipul acesteia. De exemplu, dacă 
expresia de selecŃie este de tip std_logic_vector de 2 biŃi, trebuie enumerate 81 de 
valori, deoarece pentru un singur bit pot exista 9 valori posibile. Dacă nu sunt 
enumerate toate valorile, trebuie utilizată clauza others. 

• Clauza others trebuie să fie ultima dintre toate opŃiunile. 
 

Exemplul 1 prezintă un proces pentru secvenŃierea valorilor unui tip enumerat 
reprezentând stările unui semafor. Procesul este descris cu ajutorul unei instrucŃiuni case. 

Exemplul 1: 
type tip_culoare is (rosu, galben, verde); 
process (culoare) 

case culoare is 
when rosu => 

culoare_urm <= verde; 
when galben => 

culoare_urm <= rosu; 
when verde => 

culoare_urm <= galben; 
end case; 

end process; 
 

Exemplul 2 defineşte o entitate şi o arhitectură pentru o poartă SAU EXCLUSIV cu 
două intrări. Poarta este descrisă în mod funcŃional cu ajutorul unui proces care conŃine o 
instrucŃiune case. 

Exemplul 2: 
library ieee; 
use ieee.std_logic_1164.all; 
entity xor2 is 

port (a, b: in std_logic; 
x: out std_logic); 

end xor2; 
architecture arh_xor2 of xor2 is 
begin 
process (a, b) 
variable temp: std_logic_vector (1 downto 0); 
begin 
temp := a & b; 

case temp is 
when "00" => x <= '0'; 
when "01" => x <= '1'; 
when "10" => x <= '1'; 
when "11" => x <= '0'; 
when others => x <= '0'; 

end case; 
end process; 
end arh_xor2; 

 
4.1.2. InstrucŃiuni de buclare 

 
InstrucŃiunile de buclare permit execuŃia repetată a unei secvenŃe de instrucŃiuni, de 

exemplu, prelucrarea fiecărui element al unui tablou în acelaşi mod. În limbajul VHDL, există 
trei tipuri de instrucŃiuni de buclare: instrucŃiunea de buclare simplă loop, instrucŃiunea while 
loop şi instrucŃiunea for loop.  

InstrucŃiunea de buclare simplă loop specifică repetarea unor instrucŃiuni în mod 
nedefinit. În cazul instrucŃiunii while loop, instrucŃiunile care formează corpul buclei sunt 
repetate cât timp condiŃia specificată este adevărată, iar în cazul instrucŃiunii for loop 
instrucŃiunile din corpul buclei sunt repetate de un număr de ori specificat de un contor. 

ObservaŃie 
• Singura instrucŃiune de buclare care poate fi utilizată pentru sinteza logică este 

instrucŃiunea for loop, deoarece la aceasta numărul de iteraŃii este fix. 
 

4.1.2.1. InstrucŃiunea loop 
 

InstrucŃiunea de buclare simplă loop are sintaxa următoare: 
[etichetă:] loop 

secvenŃă_de_instrucŃiuni 

end loop [etichetă]; 
 

InstrucŃiunea are o etichetă opŃională, care poate fi utilizată pentru identificarea 
instrucŃiunii. InstrucŃiunea loop are ca efect repetarea instrucŃiunilor din corpul buclei de un 
număr nelimitat de ori. În cazul acestei instrucŃiuni, singura posibilitate de terminare a 
execuŃiei este utilizarea unei instrucŃiuni exit. 
 

4.1.2.2. InstrucŃiunea while loop 
 
InstrucŃiunea while loop este o instrucŃiune de buclare condiŃionată. Sintaxa acestei 

instrucŃiuni este următoarea: 
[etichetă:] while condiŃie loop 

secvenŃă_de_instrucŃiuni 

end loop [etichetă]; 
 

CondiŃia este testată înaintea fiecărei execuŃii a buclei. Dacă această condiŃie este 
adevărată, se execută secvenŃa de instrucŃiuni din corpul buclei, după care controlul este 
transferat la începutul buclei. ExecuŃia buclei se termină atunci când condiŃia testată devine 
falsă, caz în care se execută instrucŃiunea care urmează după clauza end loop. 

Exemplul 3: 
process 

variable contor: integer := 0; 
begin 

wait until clk = '1'; 
while nivel = '1' loop 

contor := contor + 1; 
wait until clk = '0'; 
end loop; 

end process; 
 

În procesul din Exemplul 3 se numără fronturile crescătoare ale semnalului de ceas clk 
atât timp cât semnalul nivel are valoarea '1'. Stabilitatea semnalului nivel nu este testată. Se 
testează doar dacă acest semnal are valoarea '1' la detectarea unui front crescător al semnalului 
clk, şi nu se testează dacă valoarea semnalului nivel s-a modificat de la testarea anterioară. 

Corpul buclei unei instrucŃiuni while loop poate conŃine o altă instrucŃiune de buclare, 
în particular o instrucŃiune while loop, după cum se arată în exemplul următor. 

E1: while i < 10 loop 
E2: while j < 20 loop 
... 
end loop E2; 
end loop E1; 

 
4.1.2.3. InstrucŃiunea for loop 

 
Pentru execuŃia repetată a unei secvenŃe de instrucŃiuni de un număr de ori specificat 

se poate utiliza instrucŃiunea for loop. Sintaxa acestei instrucŃiuni este următoarea: 
[etichetă:] for contor in domeniu loop 

secvenŃă_de_instrucŃiuni 

end loop [etichetă]; 
 

Într-o instrucŃiune for loop se specifică un contor de iteraŃii şi un domeniu. 
InstrucŃiunile din corpul buclei sunt executate cât timp contorul se află în domeniul specificat. 
După terminarea unei iteraŃii, contorului i se asignează următoarea valoare din domeniu. 
Domeniul poate fi unul crescător, specificat prin cuvântul cheie to, sau descrescător, 
specificat prin cuvântul cheie downto. Acest domeniu poate fi specificat şi sub forma unui tip 
enumerat sau subtip, caz în care nu se specifică în mod explicit limitele domeniului în cadrul 
instrucŃiunii for loop. Limitele domeniului vor fi determinate de compilator din declaraŃia 
tipului sau subtipului respectiv. 

InstrucŃiunea for loop din Exemplul 4 calculează pătratele valorilor întregi cuprinse 
între 1 şi 10,  pe care le depune în tabloul i_patrat. 

Exemplul 4: 
for i in 1 to 10 loop 

i_patrat (i) <= i * i; 
end loop; 

 
Contorul de iteraŃii din acest exemplu este în mod implicit de tip integer, deoarece 

tipul acestuia nu a fost definit în mod explicit. Forma completă a declaraŃiei domeniului 
pentru contorul de iteraŃii este similară cu cea a unui tip. Pentru exemplul anterior, clauza for 
poate fi scrisă şi sub forma următoare: 

for i in integer range 1 to 10 loop 
 

În unele limbaje de programare, în cadrul buclei se poate asigna o valoare contorului 
de iteraŃii (în exemplul anterior, i) pentru a-i modifica valoarea. Limbajul VHDL nu permite 
însă asignarea unei valori contorului de iteraŃii sau utilizarea acestuia ca parametru de intrare 
sau de ieşire al unei proceduri. Contorul poate fi utilizat însă într-o expresie, cu condiŃia să nu 
i se modifice valoarea. Un alt aspect legat de contorul de iteraŃii este că acesta nu trebuie 
declarat în mod explicit în cadrul procesului. Acest contor este declarat în mod implicit ca o 
variabilă locală a instrucŃiunii de buclare prin specificarea sa după cuvântul cheie for. Dacă 
există o altă ariabilă cu acelaşi nume în cadrul procesului, cele două vor fi tratate ca variabile 
separate. 

Interpretarea sintezei instrucŃiunii for loop este aceea că se realizează o nouă copie a 
circuitului descris de conŃinutul instrucŃiunii la fiecare iteraŃie a buclei. Utilizarea instrucŃiunii 
for loop pentru generarea unui circuit este ilustrată în Exemplul 5. 
Exemplul 5: 

entity potrivire_biti is 
port (a, b: in bit_vector (7 downto 0); 

potriviri: out bit_vector (7 downto 0)); 
end potrivire_biti; 
architecture functional of potrivire_biti is 
begin 
process (a, b) 
begin 

for i in 7 downto 0 loop 
potriviri (i) <= not (a(i) xor b(i)); 

end loop; 
end process; 
end functional; 

 
Prin procesul din acest exemplu, se generează un set de comparatoare de câte un bit, 

care compară biŃii de acelaşi rang ai vectorilor a şi b. Rezultatul este înscris în vectorul 
potriviri, care va conŃine '1' în poziŃiile în care biŃii celor doi vectori sunt identici şi '0' în 
celelalte poziŃii. Procesul din exemplul anterior este echivalent cu următorul proces: 

process (a, b) 
begin 

potriviri (7) <= not (a(7) xor b(7)); 
potriviri (6) <= not (a(6) xor b(6)); 
potriviri (5) <= not (a(5) xor b(5)); 
potriviri (4) <= not (a(4) xor b(4)); 
potriviri (3) <= not (a(3) xor b(3)); 
potriviri (2) <= not (a(2) xor b(2)); 
potriviri (1) <= not (a(1) xor b(1)); 

potriviri (0) <= not (a(0) xor b(0)); 
end process; 

 
În acest caz, ordonarea domeniului contorului de iteraŃii nu are importanŃă, deoarece 

nu există conexiune între blocurile generate ale circuitului. Această ordonare devine 
importantă atunci când există o conexiune între aceste blocuri. Această conexiune este creată 
de obicei de către o variabilă care păstrează o valoare într-o iteraŃie a buclei, valoare care este 
citită apoi într-o altă iteraŃie. De obicei, este necesară iniŃializarea unei asemene variabile 
înaintea intrării în buclă. Exemplul 6 prezintă un asemenea circuit. 

Exemplul 6: 
library ieee; 
use ieee.numeric_bit.all; 
entity contorizare_unu is 

port (v: in bit_vector (15 downto 0); 
num: out signed (3 downto 0)); 

end contorizare_unu; 
architecture functional of contorizare_unu is 
begin 
process (v) 

variable rez: signed (3 downto 0); 
begin 

rez := (others => '0'); 
for i in 15 downto 0 loop 

if v(i) = '1' then 
rez := rez + 1; 

end if; 
end loop; 

num <= rez; 
end process; 
end functional; 

 
Acest exemplu este un circuit combinaŃional care contorizează numărul biŃilor din 

vectorul v care sunt '1'. Rezultatul este acumulat pe parcursul execuŃiei procesului într-o 
variabilă numită rez şi este asignată apoi semnalului de ieşire num la sfârşitul procesului.  

În Exemplul 6, domeniul contorului de iteraŃii al instrucŃiunii for loop a fost specificat 
în mod explicit ca fiind 15 downto 0. În practică, această specificare explicită este utilizată 
foarte rar pentru accesarea tablourilor, fiind recomandată utilizarea atributelor pentru tablouri.  

 
4.1.2.4. InstrucŃiunea next 

 
Există situaŃii în care este necesară oprirea execuŃiei instrucŃiunilor dintr-o buclă în 

iteraŃia curentă şi trecerea la următoarea iteraŃie. Pentru aceasta se poate utiliza instrucŃiunea 
next. Sintaxa acestei instrucŃiuni este următoarea: 

next [etichetă] [when condiŃie]; 
 

La întâlnirea unei instrucŃiuni next în cadrul corpului unei bucle, execuŃia iteraŃiei 
curente este oprită şi controlul este transferat la începutul instrucŃiunii de buclare, fie 
necondiŃionat, dacă clauza when nu este prezentă, fie condiŃionat, dacă această clauză este 
prezentă. Contorul de iteraŃii va fi actualizat, iar dacă limita domeniului nu a fost atinsă, 

execuŃia va continua cu prima instrucŃiune din corpul buclei. În caz contrar, execuŃia 
instrucŃiunii de buclare se va termina. 

În cazul în care există mai multe nivele de instrucŃiuni de buclare (o buclă conŃinută 
într-o altă buclă), în cadrul instrucŃiunii next se poate specifica o etichetă, aceasta fiind 
eticheta instrucŃiunii de buclare de nivel imediat superior în care este cuprinsă instrucŃiunea 
next. Această etichetă are doar rolul de a creşte claritatea descrierii şi nu poate fi diferită de 
cea a instrucŃiunii de buclare curente. 

O instrucŃiune next se poate utiliza în locul unei instrucŃiuni if pentru execuŃia 
condiŃionată a unui grup de instrucŃiuni. Pentru sinteza instrucŃiunii next sunt necesare 
aceleaşi circuite ca şi cele necesare pentru sinteza instrucŃiunii if. Alegerea uneia din cele 
două instrucŃiuni rămâne la latitudinea proiectantului. 

Ca un exemplu, considerăm acelaşi circuit care contorizează numărul biŃilor de '1' 
dintr-un vector. Exemplul 7 prezintă descrierea modificată a acestui circuit. În locul 
instrucŃiunii if se utilizează o instrucŃiune next, prin care se abandonează o iteraŃie în cazul în 
care valoarea elementului curent este '0', astfel că nu se execută incrementarea contorului. 

Exemplul 7: 
library ieee; 
use ieee.numeric_bit.all; 
entity contorizare_unu is 

port (v: in bit_vector (15 downto 0); 
num: out signed (3 downto 0)); 

end contorizare_unu; 
architecture functional of contorizare_unu is 
begin 
process (v) 

variable rez: signed (3 downto 0); 
begin 

rez := (others => '0'); 
for i in v'range loop 

next when v(i) = '0'; 
rez := rez + 1; 

end loop; 
num <= rez; 

end process; 
end functional; 

 
4.1.2.5. InstrucŃiunea exit 

 
Există situaŃii în care execuŃia unei instrucŃiuni de buclare trebuie oprită complet, fie 

datorită apariŃiei unei erori în timpul execuŃiei unui model, fie datorită faptului că prelucrarea 
trebuie terminată înaintea depăşirii domeniului de către contorul de iteraŃii. În asemenea 
situaŃii, se poate utiliza instrucŃiunea exit. Sintaxa acestei instrucŃiuni este următoarea: 

exit [etichetă] [when condiŃie]; 
 

Există trei forme ale instrucŃiunii exit. Prima este cea în care nu apare o etichetă şi nici 
o condiŃie specificată printr-o clauză when. În acest caz, se va opri în mod necondiŃionat 
execuŃia instrucŃiunii curente de buclare. 

În cazul în care instrucŃiunea exit apare într-o instrucŃiune de buclare aflată în 
interiorul unei alte instrucŃiuni de buclare, va fi oprită doar execuŃia instrucŃiunii interioare de 
buclare, dar execuŃia instrucŃiunii exterioare de buclare va continua. 

Dacă în instrucŃiunea exit se specifică o etichetă a unei instrucŃiuni de buclare, la 
întâlnirea instrucŃiunii exit controlul va fi transferat la eticheta specificată. 

Dacă instrucŃiunea exit conŃine o clauză when, execuŃia instrucŃiunii de buclare va fi 
oprită numai în cazul în care condiŃia specificată de această clauză este adevărată. Următoarea 
instrucŃiune executată depinde de prezenŃa sau absenŃa unei etichete în cadrul instrucŃiunii. 
Dacă se specifică o etichetă a unei instrucŃiuni de buclare, următoarea instrucŃiune executată 
va fi prima din instrucŃiunea de buclare specificată de acea etichetă. Dacă nu se specifică o 
etichetă, următoarea instrucŃiune executată va fi cea de după clauza end loop a instrucŃiunii de 
buclare curente. 

InstrucŃiunea exit se poate utiliza pentru a termina execuŃia unei instrucŃiuni loop 
simple, după cum se arată în Exemplul 8. 

Exemplul 8: 
E3: loop 

a := a + 1; 
exit E3 when a > 10; 

end loop E3; 
 

Exemplul 9 prezintă descrierea unui circuit pentru contorizarea numărului de zerouri 
de la sfârşitul unui vector de biŃi. Se testează fiecare element al vectorului reprezentând o 
valoare întreagă, iar dacă un element este '1', bucla se termină cu ajutorul instrucŃiunii exit. 

Exemplul 9: 
library ieee; 
use ieee.numeric_bit.all; 
entity contorizare_zero is 

port (v: in bit_vector (15 downto 0); 
num: out signed (3 downto 0)); 

end contorizare_zero; 
architecture functional of contorizare_zero is 
begin 
process (v) 

variable rez: signed (3 downto 0); 
begin 

rez := (others => '0'); 
for i in v'reverse_range loop 

exit when v(i) = '1'; 
rez := rez + 1; 

end loop; 
num <= rez; 

end process; 
end functional; 
 
4.1.3. InstrucŃiunea secvenŃială assert 

 
InstrucŃiunea assert este utilă pentru afişarea unor mesaje de avertisment sau de eroare 

în timpul simulării unui model. Această instrucŃiune testează valoarea unei condiŃii booleene 

şi afişează mesajul specificat în cazul în care condiŃia este falsă. Sintaxa instrucŃiunii este 
următoarea: 

assert condiŃie 

[report şir_de_caractere] 
[severity nivel_de_severitate]; 

 
CondiŃia specificată este o expresie care trebuie să se evalueze la o valoare booleană. 

Dacă această valoare este TRUE, instrucŃiunea nu are nici un efect. Dacă valoarea este 
FALSE, se afişează textul specificat în clauza report. 

Clauza opŃională report poate avea ca argument un şir de caractere, cu tipul predefinit 
string. Dacă această clauză nu este specificată, şirul de caractere care va fi afişat în mod 
implicit va fi "Assertion violation". 

Clauza opŃională severity permite specificarea nivelului de severitate al violării 
aserŃiunii. Nivelul de  severitate trebuie să fie o expresie cu tipul predefinit severity_level. 
Acest tip conŃine următoarele valori, în ordinea crescătoare a nivelului de severitate: note, 
warning, error şi failure. Dacă această clauză este omisă, se va presupune nivelul de 
severitate implicit error. Utilizarea nivelelor de severitate este descrisă pe scurt în continuare. 

• Nivelul note poate fi utilizat pentru afişarea unor informaŃii despre modul de 
desfăşurare al simulării. 

• Nivelul warning poate fi utilizat în situaŃiile în care simularea poate fi continuată, dar 
este posibil ca rezultatele să fie imprevizibile. 

• Nivelul error se utilizează atunci când violarea aserŃiunii reprezintă o eroare care 
determină funcŃionarea incorectă a modelului, în acest caz execuŃia simulării fiind 
oprită. 

• Nivelul failure se utilizează atunci când violarea aserŃiunii reprezintă o eroare fatală, 
cum este împărŃirea la zero sau adresarea unui tablou cu un index care depăşeşte 
domeniul permis. Şi în acest caz, execuŃia simulării va fi oprită. 

 
Exemplul 10 
assert not (R = '1' and S = '1') 

report "Ambele semnale R şi S au valoarea '1'" 
severity error; 

 
Atunci când ambele semnale R şi S au valoarea '1', se afişează mesajul specificat şi 

simularea va fi oprită.  
Pentru afişarea unui mesaj în mod necondiŃionat, se va utiliza condiŃia FALSE, de 

exemplu: 
assert (FALSE) report "Start simulare"; 

 
ObservaŃii 

• InstrucŃiunea assert descrisă este o instrucŃiune secvenŃială, presupunând că ea apare 
într-un proces sau subprogram. Există însă şi o versiune concurentă a acestei 
instrucŃiuni. Aceasta are un format identic cu versiunea secvenŃială, dar poate apare 
numai în afara unui proces sau subprogram. 

• De obicei, sistemele de sinteză ignoră instrucŃiunea assert. 

4.3. Teme propuse 
 

4.3.1. AnalizaŃi exemplele prezentate. SimulaŃi descrierea circuitului pentru 
compararea biŃilor de acelaşi rang a doi vectori (Exemplul 5), a circuitului pentru contorizarea 
biŃilor de '1' ai unui vector (Exemplul 6). UtilizaŃi sistemul Active-HDL pentru compilarea cu 
succes a acestor descrieri. 

4.3.2. RealizaŃi un comparator de 4 biŃi cu trei ieşiri (egal, mai mic, mai mare) 
utilizând: 

a) Operatori logici; 
b) InstrucŃiunea de asignare selectivă; 
c) InstrucŃiunea condiŃională if. 

CompilaŃi descrierile comparatorului şi simulaŃi funcŃionarea acestora. 
4.3.2. DescrieŃi un multiplexor 4:1 de 8 biŃi utilizând o instrucŃiune case. Intrările 

multiplexorului sunt a[7:0], b[7:0], c[7:0], d[7:0], s[1:0], iar ieşirile sunt q[7:0]. 
4.3.3. Să se realizeze o descriere a unui decodificator pe 3 biti, cu o intrare de 3 biti, 

intrare de activare a iesirii si 8 iesiri de câte 1 bit. Introducerea unui cod pe intrare generează 
activarea iesirii corespunzătoare codului aplicat. 

• dacă In=0 atunci iesirea corespunzătoare codului se activează y0=1, iar celelalte iesiri 
sunt 0; 

• dacă In=1 atunci iesirea corespunzătoare codului se activează y1=1, s.a.m.d. 
4.3.4. RealizaŃi un decodificator din codul BCD pentru afişajul cu 7 segmente. Intrările 

decodificatorului sunt bcd[3:0], iar ieşirile sunt led[6:0]. 
 


